

Use these laws, together with the definition of *unzip*, to prove the required result.

4.5 The fold functions

We have seen in the case of the datatype *Nat* that many recursive definitions can be expressed very succinctly using a suitable fold operator. Exactly the same is true of lists. Consider the following definition of a function *h*:

$$\begin{aligned} h [] &= e \\ h (x : xs) &= x \oplus h xs \end{aligned}$$

The function *h* works by taking a list, replacing `[]` by *e* and `(:)` by \oplus , and evaluating the result. For example, *h* converts the list

$$x_1 : (x_2 : (x_3 : (x_4 : [])))$$

to the value

$$x_1 \oplus (x_2 \oplus (x_3 \oplus (x_4 \oplus e)))$$

Since `(:)` associates to the right, there is no need to put in parentheses in the first expression. However, we do have to put in parentheses in the second expression because we do not assume that \oplus associates to the right.

The pattern of definition given by *h* is captured in a function *foldr* (pronounced ‘fold right’) defined as follows:

$$\begin{aligned} \textit{foldr} &:: (\alpha \rightarrow \beta \rightarrow \beta) \rightarrow \beta \rightarrow [\alpha] \rightarrow \beta \\ \textit{foldr} f e [] &= e \\ \textit{foldr} f e (x : xs) &= f x (\textit{foldr} f e xs) \end{aligned}$$

We can now write $h = \textit{foldr}(\oplus)e$. The first argument of *foldr* is a binary operator that takes an α -value on its left and a β -value on its right, and delivers a β -value. The second argument of *foldr* is a β -value. The third argument is of type `[α]`, and the result is a value of type β . In many cases, α and β will be instantiated to the same type, for instance when \oplus denotes an associative operation.

The single function *foldr* can be used to define almost every function on lists that we have met so far. Here are just some examples:

$$\begin{aligned} \textit{concat} &:: [[\alpha]] \rightarrow [\alpha] \\ \textit{concat} &= \textit{foldr} (+) [] \end{aligned}$$

```

reverse  ::  [α] → [α]
reverse  =  foldr snoc [ ]
           where snoc x xs = xs ++ [x]

```

```

length   ::  [α] → Int
length   =  foldr oneplus 0
           where oneplus x n = 1 + n

```

```

sum       ::  Num α ⇒ [α] → α
sum       =  foldr (+) 0

```

```

and       ::  [Bool] → Bool
and       =  foldr (∧) True

```

```

map       ::  (α → β) → [α] → [β]
map f     =  foldr (cons · f) [ ]
           where cons x xs = x : xs

```

```

unzip     ::  [(α, β)] → ([α], [β])
unzip     =  foldr cons ([ ], [ ])
           where cons (x, y) (xs, ys) = (x : xs, y : ys)

```

In some of these definitions the operator \oplus that replaces $(:)$ is associative, and the constant c that replaces $[]$ is the unit of \oplus . For example, $(+, [])$, $(+, 0)$, (\wedge, True) , and so on. In other definitions, \oplus is not associative; indeed, \oplus may not even have a type appropriate for an associative operator, namely $\alpha \rightarrow \alpha \rightarrow \alpha$.

Not every function on lists can be defined as an instance of *foldr*. For example, *zip* cannot be so defined. Even for those that can, an alternative definition may be more efficient. To illustrate, suppose we want a function *decimal* that takes a list of digits and returns the corresponding decimal number; thus

$$\text{decimal } [x_0, x_1, \dots, x_n] = \sum_{k=0}^n x_k 10^{(n-k)}$$

It is assumed that the most significant digit comes first in the list. One way to compute *decimal* efficiently is by a process of multiplying each digit by ten and adding in the following digit. For example,

$$\text{decimal } [x_0, x_1, x_2] = 10 \times (10 \times (10 \times 0 + x_0) + x_1) + x_2$$

This decomposition of a sum of powers is known as *Horner's rule*. Suppose we define \oplus by $n \oplus x = 10 \times n + x$. Then we can rephrase the above equation as

$$\text{decimal} [x_0, x_1, x_2] = ((0 \oplus x_0) \oplus x_1) \oplus x_2$$

This is almost like an instance of *foldr*, except that the grouping is the other way round, and the starting value appears on the left, not on the right. In fact, the computation is dual: instead of processing from right to left, the computation processes from left to right.

This example motivates the introduction of a second fold operator, called *foldl* (pronounced 'fold left'). Informally,

$$\text{foldl} (\oplus) e [x_0, x_1, \dots, x_{n-1}] = (\dots ((e \oplus x_0) \oplus x_1) \dots) \oplus x_{n-1}$$

The parentheses group from the left, which is the reason for the name. The full definition of *foldl* is

$$\begin{aligned} \text{foldl} &:: (\beta \rightarrow \alpha \rightarrow \beta) \rightarrow \beta \rightarrow [\alpha] \rightarrow \beta \\ \text{foldl } f \ e [] &= e \\ \text{foldl } f \ e (x : xs) &= \text{foldl } f \ (f \ e \ x) \ xs \end{aligned}$$

For example,

$$\begin{aligned} &\text{foldl} (\oplus) e [x_0, x_1, x_2] \\ &= \text{foldl} (\oplus) (e \oplus x_0) [x_1, x_2] \\ &= \text{foldl} (\oplus) ((e \oplus x_0) \oplus x_1) [x_2] \\ &= \text{foldl} (\oplus) (((e \oplus x_0) \oplus x_1) \oplus x_2) [] \\ &= ((e \oplus x_0) \oplus x_1) \oplus x_2 \end{aligned}$$

If \oplus is associative with unit e , then *foldr* $(\oplus) e$ and *foldl* $(\oplus) e$ define the same function on finite lists, as we will see in the following section.

As another example of the use of *foldl*, consider the following definition:

$$\begin{aligned} \text{reverse}' &:: [\alpha] \rightarrow [\alpha] \\ \text{reverse}' &= \text{foldl } \text{cons} [] \quad \text{where } \text{cons } xs \ x = x : xs \end{aligned}$$

Note the order of the arguments to *cons*; we have *cons* = *flip* (:), where the standard function *flip* is defined by *flip* $f \ x \ y = f \ y \ x$. The function *reverse'* reverses a finite list. For example,

$$\begin{aligned} &\text{reverse}' [x_0, x_1, x_2] \\ &= \text{cons} (\text{cons} (\text{cons} [] x_0) x_1) x_2 \\ &= \text{cons} (\text{cons} [x_0] x_1) x_2 \\ &= \text{cons} [x_1, x_0] x_2 \\ &= [x_2, x_1, x_0] \end{aligned}$$

One can prove that $reverse' = reverse$ by induction, or as an instance of a more general result to be described in the following section. Of greater importance than the mere fact that $reverse$ can be defined in a different way, is that $reverse'$ gives a much more efficient program: $reverse'$ takes time proportional to n on a list of length n , while $reverse$ takes time proportional to n^2 .

There are a number of important laws concerning $foldr$ and its relationship to $foldl$, but we will postpone discussion to the following section.

4.5.1 Fold over nonempty lists

Say we wish to find the maximum element in a finite list of elements drawn from an ordered type. We would like to do this by defining

$$\begin{aligned} maxlist &:: (Ord\ \alpha) \Rightarrow [\alpha] \rightarrow \alpha \\ maxlist &= foldr\ (\mathbf{max})\ e \end{aligned}$$

where $(x\ \mathbf{max}\ y)$ returns the greater of x and y . But what should we choose as the value of e ? In other words, what is $maxlist\ []$? The choice is not arbitrary because we naturally want $maxlist\ [x] = x$, and the definition gives $(x\ \mathbf{max}\ e)$ instead. So e has to be smaller than any other value in the type α . Such a value may not exist. In Haskell there is a type class *Bounded* whose instances are types possessing maximum and minimum elements:

```
class Bounded  $\alpha$  where
  minBound, maxBound ::  $\alpha$ 
```

Hence one solution is to redefine $maxlist$ to

$$\begin{aligned} maxlist &:: (Ord\ \alpha, Bounded\ \alpha) \Rightarrow [\alpha] \rightarrow \alpha \\ maxlist &= foldr\ (\mathbf{max})\ minBound \end{aligned}$$

An alternative solution that does not make use of *Bounded* is to introduce a new fold function, one that works on nonempty lists. In fact, there are two such functions, $foldr1$ and $foldl1$. The definitions are

$$\begin{aligned} foldr1 &:: (\alpha \rightarrow \alpha \rightarrow \alpha) \rightarrow [\alpha] \rightarrow \alpha \\ foldr1\ f\ (x : xs) &= \text{if null } xs \text{ then } x \text{ else } f\ x\ (foldr1\ f\ xs) \\ foldl1 &:: (\alpha \rightarrow \alpha \rightarrow \alpha) \rightarrow [\alpha] \rightarrow \alpha \\ foldl1\ f\ (x : xs) &= foldl\ f\ x\ xs \end{aligned}$$

For example,

$$\begin{aligned} foldr1\ (\oplus)\ [x_0, x_1, x_2, x_3] &= x_0 \oplus (x_1 \oplus (x_2 \oplus x_3)) \\ foldl1\ (\oplus)\ [x_0, x_1, x_2, x_3] &= ((x_0 \oplus x_1) \oplus x_2) \oplus x_3 \end{aligned}$$

Now we can solve the problem by defining

$$\text{maxlist} = \text{foldr1}(\text{max})$$

Of course, since **max** is associative, we could equally well have defined *maxlist* in terms of *foldl1*.

4.5.2 Scan left

Sometimes it is convenient to apply a *foldl* operation to every initial segment of a list. This is done by a function *scanl*, pronounced ‘scan left’. For example,

$$\text{scanl}(\oplus) e [x_0, x_1, x_2] = [e, e \oplus x_0, (e \oplus x_0) \oplus x_1, ((e \oplus x_0) \oplus x_1) \oplus x_2]$$

In particular, *scanl* (+) 0 computes the list of accumulated sums of a list of numbers, and *scanl* (×) 1 [1..n] computes a list of the first *n* factorial numbers. More substantial examples of the use of *scanl* are given in Sections 5.2 and 5.3. We will give two programs for *scanl*; the first is the clearest, while the second is more efficient.

For the first program we will need the function *inits* that returns the list of all initial segments of a list. For example,

$$\text{inits} [x_0, x_1, x_2] = [[], [x_0], [x_0, x_1], [x_0, x_1, x_2]]$$

The empty list has only one segment, namely the empty list itself; a list (*x* : *xs*) has the empty list as its shortest initial segment, and all the other initial segments begin with *x* and are followed by an initial segment of *xs*. Hence

$$\begin{aligned} \text{inits} &:: [\alpha] \rightarrow [[\alpha]] \\ \text{inits} [] &= [[]] \\ \text{inits} (x : xs) &= [] : \text{map} (x :) (\text{inits} xs) \end{aligned}$$

The function *inits* can be defined more succinctly as an instance of *foldr*:

$$\text{inits} = \text{foldr} f [[]] \quad \text{where } f \ x \ xss = [] : \text{map} (x :) xss$$

Now we define

$$\begin{aligned} \text{scanl} &:: (\alpha \rightarrow \beta \rightarrow \beta) \rightarrow \beta \rightarrow [\alpha] \rightarrow [\alpha] \\ \text{scanl} f e &= \text{map} (\text{foldl} f e) \cdot \text{inits} \end{aligned}$$

This is the clearest definition of *scanl* but it leads to an inefficient program. The function *f* is applied *k* times in the evaluation of *foldl f e* on a list of length *k* and, since the initial segments of a list of length *n* are lists with lengths 0, 1, ..., *n*, the function *f* is applied about $n^2/2$ times in total.

Let us now synthesise a more efficient program. The synthesis is by an induction argument on xs so we lay out the calculation in the same way.

Case $([])$. Here we get $scanl\ f\ e\ [] = [e]$. The details are routine and are left as an exercise.

Case $(x : xs)$. We calculate

$$\begin{aligned}
 & scanl\ f\ e\ (x : xs) \\
 = & \quad \{\text{definition of } scanl\} \\
 & map\ (foldl\ f\ e)\ (inits\ (x : xs)) \\
 = & \quad \{\text{definition of } inits\} \\
 & map\ (foldl\ f\ e)\ ([] : map\ (x :) (inits\ xs)) \\
 = & \quad \{\text{second equation for } map\} \\
 & foldl\ f\ e\ [] : map\ (foldl\ f\ e)\ (map\ (x :) (inits\ xs)) \\
 = & \quad \{\text{first equation for } foldl\} \\
 & e : map\ (foldl\ f\ e)\ (map\ (x :) (inits\ xs)) \\
 = & \quad \{\text{since } map \text{ is a functor}\} \\
 & e : map\ (foldl\ f\ e \cdot (x :))\ (inits\ xs) \\
 = & \quad \{\text{claim: } foldl\ f\ e \cdot (x :) = foldl\ f\ (f\ e\ x)\} \\
 & e : map\ (foldl\ f\ (f\ e\ x))\ (inits\ xs) \\
 = & \quad \{\text{definition of } scanl\} \\
 & e : scanl\ f\ (f\ e\ x)\ xs
 \end{aligned}$$

The claim is left as a short exercise. In summary, we have derived

$$\begin{aligned}
 scanl\ f\ e\ [] &= [e] \\
 scanl\ f\ e\ (x : xs) &= e : scanl\ f\ (f\ e\ x)\ xs
 \end{aligned}$$

This program is more efficient in that the function f is applied exactly n times on a list of length n .

4.5.3 Scan right

The dual computation is given by $scanr$, defined by

$$scanr\ f\ a = map\ (foldr\ f\ a) \cdot tails$$

The function *tails* returns the tail segments of a list. For example,

$$\text{tails } [x_0, x_1, x_2] = [[x_0, x_1, x_2], [x_1, x_2], [x_2], []]$$

Note that while *inits* produces a list of initial segments in increasing order of length, *tails* produces the tail segments in decreasing order of length. We can define *tails* by

$$\begin{aligned} \text{tails } [] &= [[]] \\ \text{tails } (x : xs) &= (x : xs) : \text{tails } xs \end{aligned}$$

The corresponding efficient program for *scanr* is given by

$$\begin{aligned} \text{scanr } f \ a \ [] &= [a] \\ \text{scanr } f \ a \ (x : xs) &= f \ x \ (\text{head } ys) : ys \\ &\quad \text{where } ys = \text{scanr } f \ a \ xs \end{aligned}$$

We will leave the synthesis of the more efficient version as an exercise.

4.5.4 Scans over nonempty lists

We can also define two further scan functions, *scanl1* and *scanr1*. We have

$$\begin{aligned} \text{scanl1 } f &= \text{map } (\text{foldl1 } f) \cdot \text{inits1} \\ \text{scanr1 } f &= \text{map } (\text{foldr1 } f) \cdot \text{tails1} \end{aligned}$$

where *inits1* and *tails1* return the nonempty initial and final segments of a nonempty list. For example,

$$\text{inits1 } (x : xs) = \text{map } (x :) (\text{inits } xs)$$

Using $\text{foldl1 } f \cdot (x :) = \text{foldl } f \ x$, we obtain

$$\begin{aligned} \text{scanl1} &:: (\alpha \rightarrow \alpha \rightarrow \alpha) \rightarrow [\alpha] \rightarrow [\alpha] \\ \text{scanl1 } f \ (x : xs) &= \text{scanl } f \ x \ xs \end{aligned}$$

The efficient version of *scanr1* is left as an exercise.

Exercises

4.5.1 Define *filter p* as an instance of *foldr*.

4.5.2 Consider the two functions *takeWhile* and *dropWhile*, which were discussed in Exercise 4.3.7. Define *takeWhile* as an instance of *foldr*. Can *dropWhile* be defined as an instance of a fold function?

4.5.3 Which, if either, of the following equations is true?

$$\text{foldl } (-) \ x \ xs \quad = \quad x - \text{sum } xs$$

$$\text{foldr } (-) \ x \ xs \quad = \quad x - \text{sum } xs$$

4.5.4 Consider the following definition of a function *insert*:

$$\text{insert } x \ xs \quad = \quad \text{takeWhile } (\leq x) \ xs \ ++ \ [x] \ ++ \ \text{dropWhile } (\leq x) \ xs$$

Show that if *xs* is in nondecreasing order, then so is *insert x xs* for all *x*. Use *insert* to define a function *isort* for sorting a list into nondecreasing order. Use *foldr*.

4.5.5 The function *remdups* removes adjacent duplicates from a list. For example, *remdups* [1, 2, 2, 3, 3, 3, 1, 1] = [1, 2, 3, 1]. Define *remdups* using either *foldl* or *foldr*.

4.5.6 Given a list *xs* = [*x*₁, *x*₂, ..., *x*_{*n*}] of numbers, the sequence of successive maxima *ssm xs* is the longest subsequence [*x*_{*j*₁}, *x*_{*j*₂}, ..., *x*_{*j*_{*m*}}] such that *j*₁ = 1 and *x*_{*j*} < *x*_{*j*_{*k*}} for *j* < *j*_{*k*}. For example, the sequence of successive maxima of [3, 1, 3, 4, 9, 2, 10, 7] is [3, 4, 9, 10]. Define *ssm* in terms of *foldl*.

4.5.7 Prove that *scanl f e []* = [*e*].

4.5.8 Justify the efficient program for *scanr*.

4.5.9 What list does *scanl (/) [1 .. n]* produce?

4.5.10 The mathematical constant *e* is defined by $e = \sum_{n=0}^{\infty} 1/n!$. Write down an expression that can be used to evaluate *e* to some reasonable measure of accuracy.

4.5.11 Consider the datatype *Liste α* defined in Exercise 4.1.4. The fold function for this datatype is defined by

$$\text{folde} \quad :: \quad (\beta \rightarrow \alpha \rightarrow \beta) \rightarrow \beta \rightarrow \text{Liste } \alpha \rightarrow \beta$$

$$\text{folde } f \ e \ \text{Nil} \quad = \quad e$$

$$\text{folde } f \ e \ (\text{Snoc } xs \ x) \quad = \quad f \ (\text{folde } f \ e \ xs) \ x$$

The type assigned to *folde* is very nearly the same type as that assigned to *foldl*, the only difference being that [*α*] is replaced by *Liste α*. In fact,

$$\text{folde } f \ e \quad = \quad \text{foldl } f \ e \cdot \text{convert}$$

where *convert* :: [*α*] → *Liste α*. Define the function *convert* and prove by induction that the above equation holds.

4.5.12 Derive an efficient program for *scanr1*.

4.6 Laws of fold

There are a number of important laws concerning *foldr* and its relationship with *foldl*. As we saw in Section 3.3, instead of having to prove a property of a recursively defined function over a recursive datatype by writing down an explicit induction proof, one can often phrase the property as an instance of one of the laws of the fold operator for the datatype.

4.6.1 Duality theorems

The first three laws are called *duality theorems* and concern the relationship between *foldr* and *foldl*. The *first duality theorem* is as follows:

First duality theorem. Suppose \oplus is associative with unit e . Then

$$\text{foldr } (\oplus) e \text{ xs} = \text{foldl } (\oplus) e \text{ xs}$$

for all finite lists xs .

For example, we could have defined

$$\begin{aligned} \text{sum} &= \text{foldl } (+) 0 \\ \text{and} &= \text{foldl } (\wedge) \text{True} \\ \text{concat} &= \text{foldl } (++) [] \end{aligned}$$

However, as we will elaborate in Chapter 7, it is sometimes more efficient to implement a function using *foldl*, and sometimes more efficient to use *foldr*.

The *second duality theorem* is a generalisation of the first.

Second duality theorem. Suppose \oplus , \otimes , and e are such that for all x , y , and z we have

$$\begin{aligned} x \oplus (y \otimes z) &= (x \oplus y) \otimes z \\ x \oplus e &= e \otimes x \end{aligned}$$

In other words, \oplus and \otimes associate with each other, and e on the right of \oplus is equivalent to e on the left of \otimes . Then

$$\text{foldr } (\oplus) e \text{ xs} = \text{foldl } (\otimes) e \text{ xs}$$

for all finite lists xs .

Proof. The proof of the second duality theorem is by induction on xs .

Case $([])$. Both sides simplify to e .

Case $(x : xs)$. For the left-hand side we reason

$$\begin{aligned}
 & \text{foldr } (\oplus) e (x : xs) \\
 = & \quad \{\text{definition of foldr}\} \\
 & x \oplus \text{foldr } (\oplus) e xs \\
 = & \quad \{\text{induction hypothesis}\} \\
 & x \oplus \text{foldl } (\otimes) e xs
 \end{aligned}$$

For the right-hand side we reason

$$\begin{aligned}
 & \text{foldl } (\otimes) e (x : xs) \\
 = & \quad \{\text{definition of foldl}\} \\
 & \text{foldl } (\otimes) (e \otimes x) xs \\
 = & \quad \{\text{assumption}\} \\
 & \text{foldl } (\otimes) (x \oplus e) xs
 \end{aligned}$$

The two sides have simplified to two different expressions. To show that they are equal we have to resort to a second induction proof. However, if one takes as the second induction hypothesis the obvious assertion that

$$x \oplus \text{foldl } (\otimes) e xs = \text{foldl } (\otimes) (x \oplus e) xs$$

for all x and finite lists xs , then the induction doesn't work. The only way to appreciate this fact is to try it, a task we leave as an exercise. What is needed is a *stronger* hypothesis, namely

$$x \oplus \text{foldl } (\otimes) y xs = \text{foldl } (\otimes) (x \oplus y) xs$$

for all x, y , and finite lists xs . As we saw in Section 4.2 in the discussion of *reverse*, making an induction hypothesis stronger can often simplify a proof.

Case $([])$. Both sides simplify to $x \oplus a$.

Case $(z : xs)$. (Note the use of z to avoid confusion with the variable x already present in the assertion.) For the left-hand side we reason

$$x \oplus \text{foldl } (\otimes) y (z; xs)$$

$$\begin{aligned}
&= \quad \{\text{definition}\} \\
&\quad x \oplus \text{foldl } (\oplus) (y \otimes z) \text{ xs} \\
&= \quad \{\text{induction hypothesis}\} \\
&\quad \text{foldl } (\oplus) (x \oplus (y \otimes z)) \text{ xs} \\
&= \quad \{\text{assumption}\} \\
&\quad \text{foldl } (\oplus) ((x \oplus y) \otimes z) \text{ xs}
\end{aligned}$$

For the right-hand side we reason

$$\begin{aligned}
&\quad \text{foldl } (\oplus) (x \oplus y) (z : \text{xs}) \\
&= \quad \{\text{definition}\} \\
&\quad \text{foldl } (\oplus) ((x \oplus y) \otimes z) \text{ xs}
\end{aligned}$$

□

The second duality theorem has the first duality theorem as a special case, namely when $(\oplus) = (\otimes)$.

To illustrate the second duality theorem, consider the following definitions of *length* and *reverse*:

$$\begin{aligned}
\text{length} &= \text{foldr oneplus } 0, & \text{where } \text{oneplus } x \ n &= 1 + n \\
\text{length} &= \text{foldl plusone } 0, & \text{where } \text{plusone } n \ x &= n + 1 \\
\text{reverse} &= \text{foldr snoc } [], & \text{where } \text{snoc } x \ \text{xs} &= \text{xs} ++ [x] \\
\text{reverse} &= \text{foldl cons } [], & \text{where } \text{cons } \text{xs } x &= x : \text{xs}
\end{aligned}$$

The functions *oneplus*, *plusone*, and 0 meet the conditions of the second duality theorem, as do *snoc*, *cons*, and $[]$. We leave the verification as an exercise. Hence the two definitions of *length* and *reverse* are equivalent on all finite lists. It is not obvious whether there is any practical difference between the two definitions of *length*, but the second program for *reverse* is the more efficient of the two,

The proof of the *third duality theorem* is left as an exercise.

Third duality theorem. For all finite lists *xs*,

$$\text{foldr } f \ e \ \text{xs} = \text{foldl } (\text{flip } f) \ e \ (\text{reverse } \text{xs})$$

where $\text{flip } f \ x \ y = f \ y \ x$.

To illustrate the third duality theorem, consider

$$\text{foldr } (:) \ [] \ \text{xs} = \text{foldl } (\text{flip } (:)) \ [] \ (\text{reverse } \text{xs})$$

Since $\text{foldr } (:) [] = \text{id}$ and $\text{foldl } (\text{flip } (:)) [] = \text{reverse}$, we obtain

$$xs = \text{reverse } (\text{reverse } xs)$$

for all finite lists xs , a result we have already proved directly.

4.6.2 Fusion

The duality theorems deal with the relationship between the two kinds of fold operator on lists; the fusion theorems deal with one kind of fold operator only. The fusion theorems for foldr and foldl are as follows.

Fusion theorem for foldr . If f is strict, $f a = b$, and $f (g x y) = h x (f y)$ for all x and y , then

$$f \cdot \text{foldr } g a = \text{foldr } h b$$

Fusion theorem for foldl . If f is strict, $f a = b$, and $f (g x y) = h (f x) y$ for all x and y , then

$$f \cdot \text{foldl } g a = \text{foldl } h b$$

The proof of the fusion theorem for foldr is very similar to the one in the previous chapter for foldn , and is left as an exercise. The proof of the fusion theorem for foldl is trickier, and is explored in Exercise 4.6.3.

The fusion law is very general, and there are useful special cases. Four of these are considered next.

4.6.3 Fold-map fusion

Recall that $\text{map } g = \text{foldr } (\text{cons} \cdot g) []$, where $\text{cons } x xs = x : xs$. Consider what conditions ensure that

$$\text{foldr } f a \cdot \text{map } g = \text{foldr } h b$$

Looking at the conditions of the fusion theorem for foldr , we have that $\text{foldr } f a$ is a strict function, and the second condition is satisfied by taking $a = b$. The third condition expands to

$$\begin{aligned} & \text{foldr } f a ((\text{cons} \cdot g) x xs) = h x (\text{foldr } f a xs) \\ \equiv & \quad \{\text{definition of } \text{cons}\} \\ & \text{foldr } f a (g x : xs) = h x (\text{foldr } f a xs) \\ \equiv & \quad \{\text{definition of } \text{foldr}\} \end{aligned}$$

$$\begin{aligned}
 & f (g x) (foldr f a xs) = h x (foldr f a xs) \\
 \Leftarrow & \quad \{\text{generalisation}\} \\
 & f (g x) y = h x y
 \end{aligned}$$

The symbol \Leftarrow is pronounced ‘follows from’. Since $h x y = f (g x) y$ if and only if $h x = f (g x)$ if and only if $h = f \cdot g$, we get the result that

$$foldr f a \cdot map g = foldr (f \cdot g) a$$

In words, a fold after a map can always be expressed as a single fold. We will call this law the *fold-map fusion* law.

4.6.4 Fold-concat fusion

As a second example, recall that $concat = foldr (+) []$. Consider what conditions ensure that

$$foldr f a \cdot concat = foldr h b$$

Again, looking at the conditions for fusion, we have that $foldr f a$ is strict, and the second condition is satisfied by taking $a = b$. The third condition expands to

$$\begin{aligned}
 & foldr f a (xs ++ ys) = h xs (foldr f a ys) \\
 \equiv & \quad \{\text{claim: } foldr f a (xs ++ ys) = foldr f (foldr f a ys) xs\} \\
 & foldr f (foldr f a ys) xs = h xs (foldr f a ys)
 \end{aligned}$$

Hence we can take $h xs y = foldr f y xs$ or, more shortly, $h = flip (foldr f)$. The proof of the claim is left as an exercise. Hence we have shown that

$$foldr f a \cdot concat = foldr (flip (foldr f)) a$$

We will call this the *fold-concat fusion* law.

4.6.5 Bookkeeping law

As a third example, consider what conditions on f and a ensure that

$$foldr f a \cdot concat = foldr f a \cdot map (foldr f a)$$

We will call this particular equation the *bookkeeping* law. The special instance $sum \cdot concat = sum \cdot map sum$ is exploited by every bookkeeper who sums each individual day’s accounts, and then gets the year’s total by summing the sums.

We can use the fold-concat and fold-map fusion laws to rewrite the bookkeeping law in the form

$$\text{foldr} (\text{flip} (\text{foldr} f)) a = \text{foldr} (f \cdot \text{foldr} f a) a$$

This equation holds if $\text{flip} (\text{foldr} f) = f \cdot \text{foldr} f a$. Applying both sides to xs and then y , we get

$$\text{foldr} f y xs = f (\text{foldr} f a xs) y$$

This equation holds if f is associative with unit a , a proof we will leave as an exercise.

In summary, the bookkeeping law is valid whenever the arguments to foldr are an associative operation and its unit. For example, since $\text{sum} = \text{foldr} (+) 0$, the bookkeeper's accounting method is justified. As another example,

$$\text{concat} \cdot \text{concat} = \text{concat} \cdot \text{map concat}$$

This equation says that flattening a list of lists of lists into one long list can be done by concatenating outside-in or inside-out.

4.6.6 Fold-scan fusion

Consider the equation

$$1 + x_0 + x_0 \times x_1 + x_0 \times x_1 \times x_2 = 1 + x_0 \times (1 + x_1 \times (1 + x_2))$$

The equation generalises to an arbitrary list of numbers and can be expressed in the form

$$\text{foldr1} (+) \cdot \text{scanl} (\times) 1 = \text{foldr} (\odot) 1$$

where $x \odot y = 1 + x \times y$. This leads to a more general question: when does the law

$$\text{foldr1} (\oplus) \cdot \text{scanl} (\otimes) e = \text{foldr} (\odot) e$$

hold, where $x \odot y = e \oplus (x \otimes y)$?

Under the assumption that \otimes is associative and e is the unit of \otimes , we can express scanl as an instance of foldr :

$$\text{scanl} (\otimes) e = \text{foldr} g [e] \text{ where } g x xs = e : \text{map} (x \otimes) xs$$

The proof is left as an exercise. Having rewritten scanl as an instance of foldr , we can see that the required law is yet another instance of the fusion law for

foldr. The first two conditions are immediate, and the third expands to

$$\begin{aligned}
 & \text{foldr1 } (\oplus) (g \times xs) = x \odot (\text{foldr1 } (\oplus) xs) \\
 \equiv & \quad \{\text{definition of } g\} \\
 & \text{foldr1 } (\oplus) (e : \text{map } (x \otimes) xs) = x \odot (\text{foldr1 } (\oplus) xs) \\
 \equiv & \quad \{\text{definition of } \text{foldr1} \text{ and } \odot\} \\
 & e \oplus \text{foldr1 } (\oplus) (\text{map } (x \otimes) xs) = e \oplus (x \otimes \text{foldr1 } (\oplus) xs) \\
 \Leftarrow & \quad \{\text{substitution}\} \\
 & \text{foldr1 } (\oplus) \cdot \text{map } (x \otimes) = (x \otimes) \cdot \text{foldr1 } (\oplus)
 \end{aligned}$$

The last equation holds if \otimes distributes over \oplus , that is, if

$$x \otimes (y \oplus z) = (x \otimes y) \oplus (x \otimes z)$$

for all x, y , and z . The proof is left as an exercise.

In summary, if \otimes is associative with unit e , and \otimes distributes over \oplus , then $\text{foldr1 } (\oplus) \cdot \text{scanl } (\otimes) e = \text{foldr } (\odot) e$ where $x \odot y = e \oplus (x \otimes y)$. We will call this the *fold-scan* fusion law.

In the remainder of this section we will give an example of how these laws can be used in the synthesis of efficient programs.

4.6.7 Example: the maximum segment sum

The maximum segment sum problem is a famous one and its history is described in J. Bentley's *Programming Pearls* (1987). Given a sequence of numbers it is required to compute the maximum of the sums of all segments in the sequence. For example, the sequence

$$[-1, 2, -3, 5, -2, 1, 3, -2, -2, -3, 6]$$

has maximum sum 7, the sum of the segment $[5, -2, 1, 3]$. On the other hand, the sequence $[-1, -2, -3]$ has a maximum segment sum of zero, since the empty sequence $[]$ is a segment of every list and its sum is zero. It follows that the maximum segment sum is always nonnegative. (The exercises deal with a variant problem in which the empty sequence is excluded, so the maximum segment sum can be negative.)

The problem can be specified formally by

$$\begin{aligned}
 \text{mss} &:: [\text{Int}] \rightarrow \text{Int} \\
 \text{mss} &= \text{maxlist} \cdot \text{map sum} \cdot \text{segs}
 \end{aligned}$$

where *segs* returns a list of all segments of a list. This function can be defined in a number of ways, including

$$\text{segs} = \text{concat} \cdot \text{map inits} \cdot \text{tails}$$

This definition corresponds to the process of taking all the initial segments of all the tail segments. For example,

$$\text{segs "abc"} = ["", "a", "ab", "abc", "", "b", "bc", "", "c", ""]$$

The empty sequence appears four times in this list, once for every tail segment of "abc".

Direct evaluation of *mss* will take a number of steps proportional to n^3 on a list of length n . There are about n^2 segments, and summing them all takes about n^3 steps. It is not obvious that we can do better than cubic time for this problem.

However, let us see what we can achieve with some calculation. Installing the definition of *segs* in *mss*, we have to compute

$$\text{maxlist} \cdot \text{map sum} \cdot \text{concat} \cdot \text{map inits} \cdot \text{tails}$$

The only thing we can do with this expression (apart from going backwards) is to use the law $\text{map } f \cdot \text{concat} = \text{concat} \cdot \text{map} (\text{map } f)$ to replace the second and third terms, obtaining

$$\text{maxlist} \cdot \text{concat} \cdot \text{map} (\text{map sum}) \cdot \text{map inits} \cdot \text{tails}$$

There are two things we can do now: we can either apply the bookkeeping law to the first two terms, or apply the law $\text{map} (f \cdot g) = \text{map } f \cdot \text{map } g$ to the third and fourth terms. Of course, we can also do both these steps at once, in which case the result will be

$$\text{maxlist} \cdot \text{map maxlist} \cdot \text{map} (\text{map sum} \cdot \text{inits}) \cdot \text{tails}$$

For the next step we can apply the *map* law once more and obtain

$$\text{maxlist} \cdot \text{map} (\text{maxlist} \cdot \text{map sum} \cdot \text{inits}) \cdot \text{tails}$$

The term $\text{map sum} \cdot \text{inits}$ can be replaced by a *scanl*, a substitution that could have been done one step earlier:

$$\text{maxlist} \cdot \text{map} (\text{maxlist} \cdot \text{scanl } (+) 0) \cdot \text{tails}$$

Since $\text{maxlist} = \text{foldr1 } (\text{max})$ and $(+)$ distributes over (max) , the fold-scan fusion law now gives

$$\text{maxlist} \cdot \text{map} (\text{foldr } (\odot) 0) \cdot \text{tails}$$

where $x \odot y = 0 \max (x + y)$. Finally, the last two terms can be replaced by a *scanr*, and we obtain

$$mss = maxlist \cdot scanr (\odot) 0$$

The result is a linear-time program for the maximum segment sum.

Exercises

4.6.1 Prove the third duality theorem.

4.6.2 Prove the fusion theorem for *foldr*.

4.6.3 The fusion theorem for *foldl* states that $f \cdot foldl\ g\ a = foldl\ h\ b$ provided f is strict, $f\ a = b$, and $f\ (g\ x\ y) = h\ (f\ x)\ y$ for all x and y .

In the induction proof the case $(x : xs)$ requires us to show that

$$f\ (foldl\ g\ (g\ a\ x)\ xs) = foldl\ h\ (h\ (f\ a)\ x)\ xs$$

To do this we need a second induction hypothesis, namely that

$$f\ (foldl\ g\ (g\ x\ y)\ zs) = foldl\ h\ (h\ (f\ x)\ y)\ zs$$

for all x and y and for all lists zs . Prove this equation and complete the proof of the fusion law.

4.6.4 Prove the equations

$$foldr\ f\ a\ (xs ++ ys) = foldr\ f\ (foldr\ f\ a\ ys)\ xs$$

$$foldl\ f\ a\ (xs ++ ys) = foldl\ f\ (foldl\ f\ a\ xs)\ ys$$

What restrictions on xs and ys are necessary, if any?

4.6.5 Prove that $scanl\ (\otimes)\ e$ can be written as a *foldr* if \otimes is associative with unit e . Can $scanl\ (\otimes)\ e$ be written as a *foldl* without any assumptions on \otimes and e ?

4.6.6 Prove that if \otimes distributes over \oplus , then

$$foldr1\ (\oplus) \cdot map\ (x \otimes) = (x \otimes) \cdot foldr1\ (\oplus)$$

4.6.7 It is tempting to try to push the calculation of the maximum segment sum one step further and use the fold-scan fusion law to replace $maxlist \cdot scanr\ (\odot)\ 0$ by a *foldl*. Why doesn't this work? (See Exercise 4.6.10 for a version of fold-scan fusion that does work.)

4.6.8 Give a fusion law for *foldr1* and use it to give conditions on *f*, *g*, and *h* so that the law

$$\text{foldr1 } f \cdot \text{scanl1 } g = \text{foldr1 } h$$

is valid.

4.6.9 Define *segs1* by *segs1* = *concat* · *map inits1* · *tails1*, where *inits1* returns the nonempty initial segments of a list, and *tails1* the nonempty tail segments. Use this version to state and solve a version of the maximum segment sum problem that leaves out empty segments.

4.6.10 There is a version of the fold-scan fusion law that does not depend on any properties of the operators involved. The law is

$$\text{foldl1 } (\oplus) \cdot \text{scanl } (\otimes) e = \text{fst} \cdot \text{foldl } (\odot) (a, a)$$

with $(x, y) \odot z = (x \oplus t, t)$, where $t = y \otimes z$. Prove this version of fold-scan fusion.

4.7 Chapter notes

Haskell provides many more standard functions on lists than we have described in this chapter. For full details, consult the Haskell standard prelude and the library of list functions.

A great deal of work has been done on automatic or machine-aided generation of induction proofs; see, for example, Boyer and Moore (1979), Gordon, Milner, and Wadsworth (1979), Paulson (1983), Martin and Nipkow (1990).

The relationship between polymorphic functions and naturality conditions is explored in Wadler (1989); see also Bird and de Moor (1997).

The fusion laws for fold operators were systematised in Malcolm (1990), Fokkinga (1992), Jeuring (1993), and Meijer (1992).

The maximum segment sum problem is described in Bentley (1987). The derivation recorded in this chapter was first given in Bird (1989).